
アプリケーション理解

Excel VBA

【目次】

文字検索	3
アンケート結果を入力する.....	3
★チェックを付ける	5

文字検索


アンケートの中から文字を検索して、チェックを付けるプログラムを作ってみましょう。

アンケート結果を入力する

以下の様な、アンケート結果を入力しましょう。


	A	B	C	D	E	F
1						
2		パソコン初歩講座 アンケート集計				
3						
4				キーワード		
5						
6		役に立った	わかりやすい	次コースも受講したい	コメント	チェック
7		4	4		4 今まで苦手だった入力がわかるようになりました。	
8		5	4		5 初心者私でも大変わかりやすかったです。	
9		5	5		5 今まで知らなかった便利なテクニックも満載で勉強になりました。	
10		2	3		2 言葉がむずかしくて、あまりピンと来ませんでした。	
11		3	5		3 わかりやすかったのですが、もっと難しいところまでやってくれると思っていました	
12						

コメントのところは、入力が長いので、1行で入りきりません。



コメントの部分を選択して、右クリックして「セルの書式設定」をクリックして、「配置」タブの「文字の制御」で「折り返して全体を表示する」をチェックして、セルの縦幅を広げてください。

では、いつものように、ボタンを張ってプログラムを組み込んでいきましょう。



開発タブの挿入アイコンを押して、左上のボタンを選択して、シートの左上を囲みます。するとダイアログボックスが表示されて、ボタン1_Click のメソッドを作成できます。

ここで新規作成を押して、以下のソースを入力します。今回は.Find メソッドを使います。

```
Sub ボタン1_Click()  
    Dim cmtRange As Range  
    Dim myRange As Range  
    Dim fstAdr As String  
  
    Set cmtRange = Range("E7:E11")  
    cmtRange.Offset(0, 1).ClearContents '一度チェックを消す  
  
    Set myRange = cmtRange.Find( _  
        what:=Range("E4").Value, lookat:=xlPart) 'キーワードで検索  
  
    If Not myRange Is Nothing Then  
        '見つけました  
        MsgBox ("発見しました")  
    Else  
        MsgBox ("ありませんでした")  
    End If  
End Sub
```

たとえば「分かる」で検索すれば「発見しました」。「次回」で検索すれば「ありませんでした」が表示されるはずです。

オブジェクト Is オブジェクト

例) myRange Is Nothing

もしも発見できなかった時には、myRange は作成されないので=(イコール)で比較することができません。そのような場合は=ではなく Is Nothing のように聞きます。

★チェックを付ける

では、今度はどこにあったのか★を付けてみましょう。

発見された時だけ、.FindNext メソッドで検索します。.FindNext は、1 つずつ検索してその場所を返します。

```
If Not myRange Is Nothing Then
    fstAdr = myRange.Address
    Do
        myRange.Offset(0, 1).Value = "★"
        Set myRange = cmtRange.FindNext(after:=myRange)
    Loop Until fstAdr = myRange.Address
Else
    MsgBox ("ありませんでした")
```

赤字のところだけ変更します。

間違えると、応答なしになってしまいますので、実行するまえにかならず上書き保存してください。

色々と検索キーワードを切り替えてやってみてください。